

# FINANCIAL ASSET PRICING AND DECISION ANALYSIS

## 1. Introduction

When we began studying decision analysis, we assumed that the criterion for determining optimal decisions is to maximize the decision-maker's expected payoff. This assumption seemed questionable when we observed that individual decision-makers are often risk averse. But then utility theory taught us to use expected utility values, rather than expected monetary values, as our general criterion for optimal decision-making.

Now let us go from individual decision-making to decision-making on behalf of a publicly-held corporation. Decision-making in a corporation may be evaluated from the perspectives of two different groups of people: the senior managers in the corporation, and the stock-holders who own the corporation. Managers and stock-holders may agree on the basic objective of maximizing corporate profits, but their attitudes towards corporate risk-bearing are often very different.

Corporate risks often generate substantial personal risks for senior managers, and they cannot divest themselves from these risks because of incentive constraints. (A firm cannot let its managers sell short its stock!) So senior managers may want to evaluate corporate risks using a risk-averse corporate utility function similar to those we have studied for individual decision-making. From this perspective, decision analysts have found that some senior managers are comfortable with an expected-utility analysis of corporate risks, sometimes assessing risk tolerances for corporate profits that may be approximately  $1/6$  of the total value of the corporation.

But the perspective of the stock-holders is often very different. Individual stock-holders may be very risk averse, but a single corporation's risks may have only a very small impact on the overall net worth of a typical well-diversified individual stock-holder. From the perspective of such well-diversified stock-holders, the goal of corporate decision-making should be to adopt strategies that, when generally understood by investors, will increase the market value of the firm's stock. To apply this criterion, we need some theory to predict how prices in the stock market are determined.

Arbitrage pricing theory tells us that any way of determining stock-market prices that does not admit arbitrage opportunities must be consistent with an expected value criterion. But whereas utility theory extended the expected value criterion by changing the way that we measure payoffs, arbitrage pricing theory extends the expected value criterion by altering the way that we measure probabilities. But the bottom line of arbitrage pricing theory is that, from the perspective of well-diversified stock-holders, a publicly-held firm should maximize an "adjusted" expected value of profits, which are computed by some "adjusted" or "market" probabilities which may be quite different from statistical frequency probabilities.

Section 2 below offers an introduction to the general ideas of arbitrage pricing theory. Section 3 shows that the well-known capital-asset pricing model is compatible with arbitrage pricing theory and gives more intuition about what these adjusted market probabilities should look like. Sections 4 and 5 then apply arbitrage pricing theory to the estimation of option values in financial markets, following the Black-Scholes theory.

## 2. Arbitrage pricing theory

In arbitrage pricing theory, we assume that there is some list of possible states of the world such that exactly one of these states will occur and the future values of financial assets will be determined by the state that occurs.

So let us begin by considering a simple two-state model in which State 1 is a state of the world with high oil prices and State 2 is a state of the world with low oil prices. Suppose that the value of some automotive company's stock will be \$80 per share next year if State 1 occurs, but it will be \$140 if State 2 occurs. Similarly, suppose that the future value of some oil company's stock will be \$140 per share next year if State 1 occurs, but it will be \$80 if State 2 occurs. Suppose also that investors can borrow and lend at a 10% annual interest rate, so that 1.1 is the annual return ratio for risk-free bonds. If these two stocks are currently selling for \$90 per share, then an arbitrage opportunity exists. For each \$180 that we borrow, we could buy one share of each of these two stocks now, and then next year (after selling the shares and repaying the debt) we would could take a sure profit of  $(80+140)-180*1.10 = \$22$  in all possible states. By borrowing more money we could make as much money as we like, with no risk to ourselves.

If the current price per share for these two stocks were each \$100, however, then such arbitrage opportunities would not exist. At this price, if we assessed a probability 0.5 for each of the two states, then the expected returns next year per dollar invested now would be  $(.5*80+.5*140)/100 = \$1.10$ , which exactly equals the cost next year of borrowing a dollar now. If every financial asset that we can buy or sell has an expected return of 10% then, no matter how we mix investments and debts, the expected value of our portfolio next year will be 10% more than its value this year. In particular, if we start with no initial investment of our own funds, then the expected value of our portfolio must be \$0 ( $=0*1.10$ ) next year. So there cannot exist any arbitrage strategy that offers positive returns in all states with zero net initial investment. In any portfolio of debts and investments, our expected returns next year

(In the above two-state example, if the current price per share for these two stocks were each more than \$100, then an arbitrage opportunity could be constructed in which we sell the stocks short now, and invest our short-sale receipts in 10% risk-free bonds. Next year, we could use the returns from these bonds to buy back the shares that we sold short, and we would have money left over in both states.)

In general, if there exists some way of assigning probabilities to the various possible states such that every financial asset offers the same expected return ratio as risk-free bonds, then arbitrage opportunities cannot exist. With each investment offering the same expected return ratio as the risk-free bonds, no portfolio can offer a higher (or lower) expected return ratio than this risk-free return ratio on the net initial investment. So there cannot exist any arbitrage strategy for generating positive returns in all states with zero net initial investment.

Now I can tell you the main result of arbitrage pricing theory. If arbitrage opportunities do not exist in a financial market then there must exist some way of assigning probabilities to the possible states such that, when we compute expected returns using this probability distribution, the expected return ratio of every financial asset is equal to the risk-free return ratio.

To illustrate this result, consider the example shown in Figure 1, which describes a simple imaginary financial market in which three stocks are traded, and the return ratios for these stocks over the next year will depend on which of four possible states of the world occurs. The table in cells B5:E7 tells us, for each stock, what its return ratio will be next year in each possible state of

the world. The return ratio for risk-free bonds is 1.10, listed in cell A1.

	A	B	C	D	E	F	G	H
1	1.1 Risk-free return ratio (\$ next year per \$1 invested now)							
2								
3	\$Returns next year per \$1 invested now							
4		State 1	State 2	State 3	State 4		Invest now	
5	Stock A	0.95	0.90	1.30	1.20		0	
6	Stock B	0.95	1.35	0.85	1.00		0	
7	Stock C	1.30	1.10	1.15	1.00		0	
8								
9	Net \$return	0	0	0	0		Goal	
10							0	
11	Shortfall	0	0	0	0	(+ is bad!)		
12								
13	SOLVER (with Options>AssumeLinearModel):							
14	Maximize G10 by changing G5:G7,G10 subject to B11:E11<=0.							
15	Select SensitivityReport when Solver finishes.							
16								
17		State 1	State 2	State 3	State 4			
18	ShadowProby	0.01504	0.43609	0.34586	0.20301			
19		ShadowE(\$Returns)						
20	Stock A	1.1						
21	Stock B	1.1						
22	Stock C	1.1						
23								
24	FORMULAS FROM RANGE A1:G22							
25	B9.	=SUMPRODUCT(B5:B7,\$G\$5:\$G\$7)-SUM(\$G\$5:\$G\$7)*\$A\$1						
26	B9 copied to B9:E9							
27	B11.	=\$G\$10-B9						
28	B11 copied to B11:E11							
29	B20.	=SUMPRODUCT(\$B\$18:\$E\$18,B5:E5)						
30	B20 copied to B20:B22							
31	Shadow probabilities (or Lagrange Multipliers) in B18:E18 are							
32	copied (with paste-special,transpose) from Solver's SensitivityRepor							
33								
34	Result: if arbitrage opportunities do not exist, then Solver must							
35	terminate with 0 optimal value in cell G10, and the Shadow Prices							
36	(or Lagrange Multipliers) in the Sensitivity Report will give us							
37	a shadow probability distribution over the states such that all							
38	assets have the same shadow-expected return ratio as risk-free bonds							

Figure 1. Shadow probabilities in an example with no arbitrage opportunities.

Given this financial data, Figure 1 shows how to use Solver to find an arbitrage opportunity if one exists. Cells G5:G7 represent the money to be invested in each of the three stocks now, in our investment strategy. (If any of these cells becomes negative, it can be

interpreted as the amount of money to be raised now by selling the corresponding stock short.) The net investment in these stocks is assumed to come out of bonds that pay the risk-free interest rate of 10%. So the formula

$$=\text{SUMPRODUCT}(B5:B7,\$G\$5:\$G\$7)-\text{SUM}(\$G\$5:\$G\$7)*\$A\$1$$

in cell B9 represents the net returns from our investments if State 1 occurs. Copying cell B9 to B9:E9 gives us cells representing the net returns from our investments in each of the four possible states. In cell G10, we enter a "goal" of returns that our investment strategy will try to achieve in all states. The shortfall from this goal in each state is computed in cells B11:E11, by entering the formula

$$=\$G\$10-B9$$

in cell B11 and then copying B11 to B11:E11. Thus, a positive shortfall in cells B11:E11 denotes a failure to achieve the goal in some state. With this spreadsheet formulation, we can now ask Solver to find the highest goal that can be achieved in all states by an investment strategy.

After Solver has been called (by the menu command sequence Tools>Solver), in the Solver dialogue box, we tell Solver to maximize the target cell G10 by changing cells G5:G7,G10 subject to the constraints B11:E11 <= 0. (It is important to include G10 among the changing cells as shown. Use the "Add" button in the Solver dialogue box to add constraints.) To take advantage of the special linear structure of this problem, which will enable Solver to compute shadow prices, we must also go to the Solver "Options" dialogue box and check the "Assume Linear Model" option (then "OK"). Then we can go back to the basic Solver dialogue box and click the "Solve" button. When Solver finishes and announces that it has found a solution, we should also select the "sensitivity report" option, which causes Solver to add a sensitivity-report page to our workbook.

For this example, Solver will report the value 0 in cell G10 as the maximum return that can be guaranteed with no net investment. (Solver may report a value of G10 slightly different from 0, such as 1.1E-06, which denotes  $1.1 \cdot 10^{-6} = 0.0000011$ , but this tiny deviation from 0 is just due to roundoff error.) Getting this output, we know that an arbitrage strategy for guaranteeing a positive return in all states with no net investment does not exist. Thus, the main result of arbitrage pricing theory tells us that there must exist some probability distribution that

makes each stock have an expected return ratio equal to the return ratio of risk-free bonds. But where can we find this probability distribution?

Microsoft Excel 5.0 Sensitivity Report						
Changing Cells						
Cell	Name	Final Value	Reduced Cost	Objective Coefficient	Allowable Increase	Allowable Decrease
\$G\$5	Stock A Invest now	0	0	0	0.05227	0.00370
\$G\$6	Stock B Invest now	0	0	0	0.08846	0.00455
\$G\$7	Stock C Invest now	0	0	0	0.05610	0.02500
\$G\$10	Goal	0	0	1	1E+30	1
Constraints						
Cell	Name	Final Value	Shadow Price	Constraint R.H. Side	Allowable Increase	Allowable Decrease
\$B\$11	Shortfall State 1	0	0.01504	0	1E+30	1E+30
\$C\$11	Shortfall State 2	0	0.43609	0	1E+30	1E+30
\$D\$11	Shortfall State 3	0	0.34586	0	1E+30	1E+30
\$E\$11	Shortfall State 4	0	0.20301	0	1E+30	1E+30

**Figure 2. Solver's Sensitivity Report for the problem in Figure 1.**

It can be found in the "sensitivity report" page that Solver added to the workbook, when we selected this option when Solver finished the optimization. When the "Assume Linear Model" option has been selected, Solver's sensitivity report includes a "shadow price" for each constraint (see Figure 2). (When the Linear Model option is not checked, Solver calls these shadow prices "Lagrange multipliers" instead.) In this problem, we have one constraint for each of the possible states, and the shadow prices of these constraints are nonnegative numbers that sum to 1. So the shadow prices generated by Solver for this problem can be interpreted as a shadow probability distribution over the set of possible states. This shadow probability distribution from the sensitivity report has been copied (and pasted-special transposed) to the range B18:E18 in Figure 1. Then the formula

$$=SUMPRODUCT(\$B\$18:\$E\$18,B5:E5)$$

in cell B20 computes the expected return ratio for Stock A under this shadow probability distribution. Copying this formula to B16:B18 we see that all three stocks have shadow-expected return ratios equal to the return ratio on risk-free bonds in this market.

This result is completely general. If you changed the returns listed in cells B5:E7 of Figure 1, or the risk-free return ratio listed in cell A1, then one of two cases will hold. Case 1 is that the Solver will find an optimal solution with 0 as the best goal that can be guaranteed in cell G10, in which case the shadow prices in the sensitivity report will form a shadow probability distribution under which all stocks offer the same expected return ratio as the risk-free bonds. Case 2 is that Solver may report "the set cell values do not converge," which means that arbitrage opportunities are possible and returns that exceed any positive goal can be guaranteed.

(If you want to see what these arbitrage strategies look like, you can add the constraint  $G10 \leq 1000$ , which tells Solver to try to stop when it finds a way to guarantee returns higher than \$1000 with no initial investment. To get an example where such arbitrage opportunities exist, you can change the value of cell B5 in Figure 1 from 0.95 to 1.19 or higher, leaving all other parameters the same.)

### 3. The Capital Asset Pricing Model

The well-known capital-asset pricing model (CAPM) teaches students that the market value of a financial asset depends not only on its expected return but also on the covariance of its return with some "market portfolio" that is held by the typical investor. (Recall that the covariance of two random variables is their correlation coefficient multiplied by their standard deviations.) The crucial insight of CAPM is that the riskiness of asset returns can have an impact on the asset's value, but only to the extent that the asset's risks are correlated with the big risks in the overall stock market that affect even the well-diversified investor.

Let me review the basic pricing formula of CAPM in a simple model with two periods, "now" and "next year". Let the random variable  $\mathbf{M}$  denote the returns next year from each dollar invested now in the well-diversified market portfolio. (You can think of this market portfolio as some broad stock-market index, such as the Dow Jones Industrial Average.) Let the random variable  $\mathbf{X}$  denote the returns next year from owning a share of some particular stock. Let  $i$  denote the annual rate of interest on risk-free bonds, so that  $1+i$  is the annual return ratio of risk-free bonds. Then CAPM asserts that the current value per share of this stock should be

$$(E(\mathbf{X}) - k * \text{Covar}(\mathbf{X}, \mathbf{M})) / (1+i)$$

where  $k$  is a market constant (denoting the required ratio of excess returns to risks) which can be computed from the formula

$$k = (E(\mathbf{M}) - (1+i)) / \text{Covar}(\mathbf{M}, \mathbf{M})$$

(With this constant, the CAPM formula correctly yields current value \$1 for a current-dollar's worth of the market portfolio.)

The CAPM pricing formula may seem inconsistent with the results of the arbitrage pricing theory that we discussed in the preceding section. But actually these two financial theories are quite compatible. The key is that arbitrage pricing theory told us that there is some probability distribution in which all assets offer the same expected return ratio, but that probability distribution may be different from the statistical probabilities that we generally use in decision analysis. Thus, to reconcile CAPM with arbitrage pricing theory, we compute some "adjusted expected values" using a different "adjusted probability distribution" that is the shadow probability distribution of arbitrage pricing theory. For any possible state of the world  $s$ , let  $P(s)$  denote the statistical probability of state  $s$ , and let  $M(s)$  denote the value of the market returns  $\mathbf{M}$  in state  $s$ . For each state  $s$ , let  $Q(s)$  be defined by the formula

$$Q(s) = P(s) * (1 - k * (M(s) - E(\mathbf{M}))).$$

These number  $Q(s)$  will satisfy the formula

$$\sum_s Q(s) = 1,$$

so they look like another probability distribution. But more importantly, when we let  $X(s)$  denotes the value of the stock returns  $\mathbf{X}$  in state  $s$ , these numbers  $Q(s)$  will give us the formula:

$$E(\mathbf{X}) - k * \text{Covar}(\mathbf{X}, \mathbf{M}) = \sum_s Q(s) * X(s).$$

Thus, these  $Q(s)$  numbers can be interpreted as adjusted probabilities, and with this interpretation the numerator of the CAPM pricing formula becomes the adjusted-expected value of the stocks returns.

These calculations are illustrated in Figure 3. Notice that the adjusted probabilities are lower than statistical probabilities in states where the market portfolio is higher than its expected value (here States 3 and 4), and the adjusted probabilities are higher than the statistical probabilities in states where the market portfolio is lower than its expected value (States 1 and 2). This difference occurs reflects that fact that risk-averse investors would appreciate extra income

more in when their other investments are doing badly than when their other investments are doing well.

	A	B	C	D	E	F
1	1.1	Risk free return ratio (\$returns next year per \$now)				
2						
3			STATE 1:	STATE 2:	STATE 3:	STATE 4:
4		Probability	0.2	0.3	0.3	0.2
5						
6		MARKET PORTFOLIO:				
7		\$Returns per \$now	0.90	1.05	1.25	1.30
8	1.1300	E(\$returns)				
9	0.0226	Covar With Mkt (=Variance)				
10	1	\$Value now, by CAPM				
11						
12	1.3274	Required ratio of excess returns to risk				
13		Adjusted proby	0.2611	0.3319	0.2522	0.1549
14						
15		STOCK X:				
16		\$Returns/share	0	5	30	20
17	14.5	E(\$returns)				
18	1.6400	Covar with Mkt				
19	11.2027	\$Value of a share now, by CAPM				
20	11.2027	\$Value of a share now, by adjusted-expected returns				
21						
22	FORMULAS FROM RANGE A1:F20					
23	A8.	=SUMPRODUCT(C7:F7,\$C\$4:\$F\$4)				
24	A9.	=COVARPR(C7:F7,\$C\$7:\$F\$7,\$C\$4:\$F\$4)				
25	A10.	=(A8-A9*\$A\$12)/\$A\$1				
26	A12.	=(A8-A1)/A9				
27	C13.	=C4*(1-\$A\$12*(C7-\$A\$8))				
28	C13 copied to C13:F13					
29	A17.	=SUMPRODUCT(C16:F16,\$C\$4:\$F\$4)				
30	A18.	=COVARPR(C16:F16,\$C\$7:\$F\$7,\$C\$4:\$F\$4)				
31	A19.	=(A17-A18*\$A\$12)/\$A\$1				
32	A20.	=SUMPRODUCT(C16:F16,\$C\$13:\$F\$13)/\$A\$1				

Figure 3. CAPM and adjusted market probabilities in a simple example.

#### 4. Brownian motion

The famous Black-Scholes option-pricing theory is a great practical application of arbitrage pricing theory. Simulation models can help you to understand and use this Black-Scholes theory. But before describing this option-pricing theory, I must tell you something about another famous mathematical model called Brownian motion which is used by the Black-Scholes

theory.

Brownian motion is named after the botanist Robert Brown (1773-1858), who looked through a microscope and saw particles of dust moving randomly around in water. Brownian motion is a mathematical model developed to describe the movement of such particles.

Mathematicians understand Brownian motion as the limit of simple random walk models where, in each short interval of time, the particle makes a small step either to the left or to the right, so that the numerical value that represents the particle's position will either increase or decrease by some small amount.

In a Brownian-motion model, we have some process randomly generating, at each point in time, a value that we may interpret as measuring the position of some particle which is wandering along a line. During any interval of time, the difference between the final value and the initial value is the Brownian motion over that time interval. The Brownian motions over disjoint time intervals are independent random variables. As the sum of many small steps, the Brownian motion over any time interval will be a Normal random variable. The mean of the Brownian motion is proportional to the length of the time interval, but the standard deviation is proportional to the square-root of the length of the time interval, and the constants of proportionality are called the drift and the volatility respectively. That is, if the drift is  $D$  and the volatility is  $V$  then, during any time interval of length  $T$  (measured in years or some other time unit), the net change in the particle's position will be a Normal random variable with mean  $D*T$  and standard deviation  $V*(T^{0.5})$ .

Figure 4 illustrates the calculations for a simple Brownian-motion model. In this case, let us say that time is being measured in years. The drift (0.18) and the volatility (0.34) of the Brownian motion are entered into cells A2 and A3 respectively. Thus, over any one-year interval, the Brownian motion will be a Normal random variable with mean 0.18 and standard deviation 0.34. Now suppose that we are interested in the Brownian motion over some other time interval, say 2 years, which has been entered into cell A4. The Brownian motion over this time interval is a Normal random variable with mean  $A2*A4 = 0.18*2 = 0.36$  and standard deviation  $A3*(A4^{.5}) = 0.34*(2^{.5}) = 0.4808$ , which are computed in cells A5 and A6 respectively. If the initial value or position of the particle is 4, then its position after 2 years will be a Normal random variable with

mean 4.36 and standard deviation 0.481. With this initial value entered in cell A7, the positional value after T years is simulated in cell A8 by the formula

$$=A7+NORMINV(RAND(),\$A\$5,\$A\$6)$$

	A	B	C	D	E	F	G	
1	Brownian motion over time T							
2	0.18	Drift						
3	0.34	Volatility						
4	2	Length of time interval T						
5	0.36	Mean of change over time T						
6	0.4808	Stdev of change over time T						
7	4	Initial value						
8	4.347	Simulated value, time T later						
9								
10	Log-Brownian motion over time T (with log-drift and							
11	log-volatility equal to the Drift and Volatility above)							
12	54.50	Initial value						
13	78.76	Simulated value, time T later						
14								
15	FORMULAS FROM RANGE A1:B13							
16	A5.	=A2*A4						
17	A6.	=A3*(A4^0.5)						
18	A8.	=A7+NORMINV(RAND(),\$A\$5,\$A\$6)						
19	A13.	=A12*EXP(NORMINV(RAND(),\$A\$5,\$A\$6))						

**Figure 4. Simulating Brownian motion over a given time interval.**

If we assumed that the price per share of stock changed over time according to a Brownian motion, then we would be assuming that the difference between the final price and the initial price during any year will be independent of the initial price at the beginning of the year. But if the price wandered down below \$1 then a \$1 price decrease would necessarily have probability 0, because negative prices are impossible! So we cannot realistically assume that prices change according to a Brownian motion.

But it can be reasonable to assume that, during any year, the ratio of the final price divided by the initial price will be independent of the initial price at the beginning of the year. This assumption is satisfied in a probability model where the natural logarithm of the price (rather than the price itself) changes according to a Brownian motion. In such a model, we say that the price is changing over time according to log-Brownian motion. In such a log-Brownian motion, the Brownian drift and volatility of the logarithm of the price may be called the log-drift and the log-

volatility of the price.

Recall that the EXP function is the inverse of the natural logarithm function LN, and the EXP function converts addition to multiplication. So over some time interval, if the natural logarithm of a price changes by the addition of some quantity  $x$ , then the price itself over that interval is being multiplied by the quantity  $\text{EXP}(x)$ . Thus, if a stock's price changes over time according to a log-Brownian motion with log-drift  $D$  and log-volatility  $V$  then, over any time interval of length  $T$ , the price at the end of the interval will be equal to the price at the beginning of the interval multiplied by a Lognormal random variable of the form  $\text{EXP}(\mathbf{X})$ , where  $\mathbf{X}$  is a Normal random variable with mean  $D*T$  and standard deviation  $V*(T^{.5})$ .

Cells A12:A13 in Figure 4 show how to simulate such log-Brownian motion. When the initial value is entered into cell A12, the value after  $T$  years of log-Brownian motion is simulated in cell A13 by the formula

$$=A12*\text{EXP}(\text{NORMINV}(\text{RAND()},\$A\$5,\$A\$6))$$

where  $A5 = A2*A4$ ,  $A6 = A3*(A4^{.5})$ ,  $A2$  is the log-drift,  $A3$  is the log-volatility, and  $A4$  is the length of the time interval  $T$ .

The log-drift and the log-volatility for such a log-Brownian motion model of price changes can be estimated from historical price data as shown in Figure 5. The range C7:C24 in Figure 5 computes the natural logarithms of past annual growth ratios of a stock price. Then the average and standard deviations of these logarithmic growth rates are computed in cells C1 and C2, to give us our estimate of the log-drift and log-volatility.

	A	B	C	D	E	F
1			0.18	Log-drift		
2			0.34	Log-volatility		
3	BOEING CO STOCK DATA					
4		Price per share				
5	Dec77	1.85	Log-growth rate	FORMULAS FROM RANGE A1:C24		
6	Dec78	4.70	0.93	C6.	=LN(B6/B5)	
7	Dec79	5.00	0.06	C6 copied to C6:C24		
8	Dec80	6.54	0.27	C1.	=AVERAGE(C5:C24)	
9	Dec81	3.33	-0.67	C2.	=STDEV(C5:C24)	
10	Dec82	5.02	0.41			
11	Dec83	6.48	0.26			
12	Dec84	8.39	0.26			
13	Dec85	11.61	0.33			
14	Dec86	11.36	-0.02			
15	Dec87	8.22	-0.32			
16	Dec88	13.47	0.49			
17	Dec89	19.79	0.38			
18	Dec90	22.69	0.14			
19	Dec91	23.88	0.05			
20	Dec92	20.06	-0.17			
21	Dec93	21.63	0.07			
22	Dec94	23.50	0.08			
23	Dec95	39.19	0.51			
24	Dec96	53.25	0.31			

Figure 5. Estimating log-drift and log-volatility from historical data.

## 5 Black-Scholes option pricing

Consider an option to buy a share of some stock for \$60 in 2 years, where the current stock price is \$54.50. Suppose that we have estimated from past data that this stock price has an annual log-drift of 0.18, and a log-volatility of 0.34. Assuming a log-Brownian motion for this stock price, cell A13 in Figure 4 simulates its future value in two years, on the option exercise date. So in the spreadsheet shown in Figure 4, the formula =MAX(A13-60,0) would simulate the value of this option on its exercise date. With this simulation model, we could easily estimate the expected value of the option at its exercise date.

But such a statistical estimate of the expected future value of the option does not tell us its value today, even if we know the return ratio for risk-free bonds. As we have seen in Sections 2 and 3, the present value of a financial asset can be found by dividing an adjusted-expected future

value by the risk-free return ratio, but this adjusted-expected value must be computed using a special adjusted probability distribution that can be very different from the statistical probability distribution that generated our past data has been drawn. From data like that of Figure 5, what we infer about this adjusted probability in which arbitrage-pricing theory is applied?

Now we are ready for the Black and Scholes's big result. From the theory of random walks and Brownian motion, they were able to show that the adjusted probabilities of arbitrage pricing theory must yield the same log-volatility of a log-Brownian motion as we get from the statistical probabilities. With this mathematical result, we are halfway there. The log-volatility from our data, 0.34 in this example, can be used to estimate the log-volatility of the log-Brownian motion in the adjusted probabilities of arbitrage pricing theory.

Now we can estimate the log-drift of the arbitrage-adjusted log-Brownian motion of the price. Remember that the arbitrage-adjusted probabilities have the property that all financial assets offer the same adjusted-expected return ratios as risk-free bonds. Let  $r$  denote the risk-free logarithmic rate of return, which is the natural logarithm of the annual return ratio on risk-free bonds. For example, if investing \$1 today in risk-free bonds will yield \$1.10 next year, then the risk-free annual logarithmic rate of return is  $r = \text{LN}(1.10) = 0.0953$ , as shown in cell A1 of Figure 6. So the log-drift of the arbitrage-adjusted log-Brownian motion must be whatever will give this price an expected annual growth ratio equal to  $\text{EXP}(r)$ . Given the log-volatility  $V$ , the log-drift  $D$  which satisfies this expected value condition is

$$D = r - 0.5*(V^2).$$

This drift is somewhat less than the risk-free rate itself, because the volatility adds an upward bias when logarithmic growth rates are transformed into growth ratios by the nonlinear EXP function.

Figure 6 illustrates how to use these results with simulation to estimate the Black-Scholes value of an option. Basic data include the logarithmic rate of return on risk-free bonds (0.0953) in cell A1, the current price of the stock (\$54.50) in cell A5, and the log-volatility of the stock (0.34) in cell A3 (which is taken from statistical data as in Figure 5). Then the arbitrage-adjusted log-drift of the stock price is computed in cell A2 of Figure 6 by the formula

$$=A1-0.5*(A3^2)$$

which yields the value 0.0375 in Figure 6. The time until the exercise date of the option is entered

into cell A4. Then the arbitrage-adjusted simulated value of the stock at the exercise date is computed in cell A6 by the formula

$$=A5*EXP(NORMINV(RAND(),A2*A4,A3*(A4^.5)))$$

Four hundred arbitration-adjusted simulated values of the stock and corresponding values of the option at the exercise date have been tabulated in the simulation table below A12 in Figure 6. Averages of these simulated values are computed in cells B8 and C8. Notice that our estimate of the adjusted-expected future stock values in cell B8 is very close to the current price times the risk-free return ratio over 2 years ( $54.50*EXP(2*0.0953) = \$65.95$ ), as required by arbitrage pricing theory. Our estimated current value of the option \$12.39 is then computed in cell E13 by dividing the average of the arbitrage-adjusted simulated option values (\$14.99 in cell C8) by the risk-free return ratio over 2 years ( $EXP(2*0.0953) = 1.1^2 = 1.21$  in cell E12).

It may seem surprising that these calculations made no use of the statistically estimated log-drift (0.18). According to CAPM, the fact that this drift was higher than the risk-free rate (0.1) is due to the fact that the risk in this stock is highly correlated with the market portfolio. But the returns from the option will also be similarly correlated with the market portfolio, and so we discount the excess statistically-expected return for the option by using the lower arbitrage-adjusted log-drift (0.0375) instead.

	A	B	C	D	E	F	G	H	I	J	
1	0.0953	Risk-free log-rate of return									
2	0.0375	Arbitrage-adjusted log-drift									
3	0.34	Log-volatility									
4	2	Length of time interval T									
5	54.50	Initial price/share of the stock									
6	49.26	Arb-simulated value at time T									
7											
8	Averages	65.94	14.99			Consider an option to buy one share of					
9						the stock, in T years, for					
10	Arb-simulated values in T years					60	dollars.				
11		Stock	Option								
12	SimTable	49.26	0		1.21	Discount factor for T years					
13	0	90.98447	30.98447		12.39	Present discounted value of option					
14	0.0025	68.82795	8.827948								
15	0.005	12.07059	0		FORMULAS FROM RANGE A1:E13						
16	0.0075	29.16568	0		A1.	=LN(1.1)					
17	0.01	104.074	44.07397		A2.	=A1-0.5*(A3^2)					
18	0.0125	113.339	53.33904		A6.	=A5*EXP(NORMINV(RAND(),A2*A4,A3*(A4^0.5)))					
19	0.015	35.06881	0		B12.	=A6					
20	0.0175	54.8562	0		C12.	=MAX(B12-F10,0)					
21	0.02	85.45986	25.45986		B8.	=AVERAGE(B13:B412)					
22	0.0225	50.58992	0		C8.	=AVERAGE(C13:C412)					
23	0.025	41.3135	0		E12.	=EXP(A4*A1)					
24	0.0275	36.97502	0		E13.	=C8/E12					

**Figure 6. Simulating the Black-Scholes option-pricing model.**

## 6. Asset pricing with constant risk-tolerant investors

The famous capital asset pricing model (CAPM) was derived under the assumption that investors care only about the mean and standard deviation of the returns from their investment portfolio, but utility theory teaches us that this assumption may not be true. In this section, we see how the basic asset-pricing model changes when we drop this assumption and assume instead that all investors have constant risk tolerance. We maintain other basic assumptions of CAPM: that investors can freely borrow and lend at some risk-free interest rate, all investors agree about the statistical probabilities of the various possible events that could affect the future returns from financial assets, and each investor's utility is derived from his or her net monetary returns.

All the assets in the stock market together form a risky market portfolio, and these risks must be shared in some way among the investors. Viewing the entire stock market as a risk-sharing partnership that includes all investors, we can apply what we have already learned about

optimal risk sharing among constant-risk-tolerant investors. When the stock market to achieves an optimal allocation of risks, all investors must share proportionally in the overall market portfolio of risky assets, and each individual investor's share must be proportional to his or her risk tolerance. So in a market equilibrium, we can assume that the prices of all risky financial assets will be such that every investor will want to hold some proportional share of all assets in the market portfolio.

Using this assumption, the only question remaining to each investor is how much of his wealth he should invest in the market portfolio and how much he should put into risk-free bonds. This question is straightforward to analyze, as illustrated by the simple example in Figure 7. The annual return ratio on risk-free bonds (assumed here to be 1.1) is entered into cell A1. Cell A2 contains the investor's risk tolerance for monetary returns next year. In this simple example, we suppose that the returns next year from the market portfolio will depend on which of four possible states will occur. The statistical probabilities for these states are listed in cells C5:F5. Then cells C8:F8 list the corresponding returns next year per dollar invested today in the market portfolio. The quantity in cell A9 represents the amount of money that this individual invests in the market portfolio; this quantity is a decision variable for the investor. Anything that he invests comes out of risk-free bonds (if he is not borrowing by issuing bonds then he is forgoing the opportunity to buy such bonds), and so his net return in State 1 from his investment in the market portfolio is computed in cell C10 by the formula

$$=(C8-\$A\$1)*A9$$

This formula is copied to cells C10:F10 to compute the net returns to his investment in the market portfolio for all possible states. These net returns are converted to utilities by the UTIL formula in cells C11:F11. Then the expected utility of his investment is calculated in cell A12 by the formula

$$=SUMPRODUCT(C11:F11,\$C\$5:\$F\$5)$$

His certainty equivalent for this investment is computed in cell A13 by the formula

$$=UINV(A12,\$A\$2)$$

	A	B	C	D	E	F
1	1.1	Risk free return ratio (\$returns next year per \$now)				
2	2000	Individual risk tolerance (\$returns next year)				
3						
4			STATE 1:	STATE 2:	STATE 3:	STATE 4:
5		Probability	0.2	0.3	0.3	0.2
6						
7		MARKET PORTFOLIO:				
8		\$Returns per \$now	0.90	1.05	1.25	1.30
9	2599	Net \$ invested now				
10		Net \$returns	-519.74	-129.93	389.80	519.74
11		Utility	-1.2968	-1.0671	-0.8229	-0.7712
12	-0.9806	EU				
13	39	CE				
14		SOLVER (1) maximizes A13 by changing A9.				
15						
16		Adjusted proby	0.2645	0.3265	0.2518	0.1573
17						
18		STOCK X:				
19		\$Returns/share	0	5	30	20
20	11.21	Computed value/share now				
21	11.21	Price now (at which demand is evaluated below)				
22	0	Net shares bought now				
23		Net \$returns	-519.74	-129.93	389.80	519.74
24		Utility	-1.2968	-1.0671	-0.8229	-0.7712
25	-0.9806	EU				
26	39	CE				
27		SOLVER (2) maximizes A26 by changing A22.				
28						
29	FORMULAS FROM RANGE A1:F27					
30	C10.	=(C8-\$A\$1)*\$A\$9		C10 copied to C10:F10		
31	C11.	=UTIL(C10,\$A\$2)		C11 copied to C11:F11		
32	A12.	=SUMPRODUCT(C11:F11,\$C\$5:\$F\$5)				
33	A13.	=UINV(A12,\$A\$2)				
34	C16.	=C5*C11/\$A\$12		C16 copied to C16:F16		
35	A20.	=SUMPRODUCT(C19:F19,\$C\$16:\$F\$16)/\$A\$1				
36	A21.	=A20				
37	C23.	=C10+(C19-\$A\$21*\$A\$1)*\$A\$22		C23 copied to C23:F23		
38	C24.	=UTIL(C23,\$A\$2)		C24 copied to C24:F24		
39	A25.	=SUMPRODUCT(C24:F24,\$C\$5:\$F\$5)				
40	A26.	=UINV(A25,\$A\$2)				
41						
42	Notes:					
43	1. The optimal investment in A9 is proportional to RT in A2.					
44	2. The adjusted probabilities in C16:F16 do not depend on RT.					
45	3. Net shares (beyond the market portfolio) of stock A					
46	will be zero only if the price equals the computed value.					

Figure 7. An asset-pricing model with constant-risk-tolerant investors.

Now we ask Solver to maximize this certainty equivalent in cell A13 by adjusting the investment quantity in cell A9. Figure 7 shows the resulting optimal investment quantity: \$2599 for an individual with risk tolerance \$2000.

It is straightforward to verify that the optimal investment quantity in cell A9 will be proportional to the risk tolerance in cell A2. For example, if we changed the risk tolerance to \$20,000, keeping all other parameters the same, then Solver would find an optimal investment quantity equal to \$25,990. The utility values generated by the optimal investment, as shown in cells C11:F11 of Figure 7, will be the same for all investors, regardless of their risk tolerance, because any change in risk tolerance will be exactly counterbalanced by the corresponding change in the optimal investment quantity.

The optimal investment quantity for each investor thus depends on the risk-free interest rate. With a lower return ratio on risk free debt in cell A1, the optimal investment in the risky market portfolio would increase for any individual investor. In a market equilibrium, the risk-free interest rate is determined by the condition that the sum of all individuals' investments in shares of the market portfolio must equal the whole portfolio of stocks and other financial assets available in the market.

So in a market equilibrium, all financial assets must be priced now in such a way that, when individual investors solve the optimal investment problem that we have just described, they buy different shares of the general market portfolio, and no one wants to buy or sell any further shares of any specific stock. The range A16:F20 in Figure 7 shows how to compute the current price per share of any specific stock that will satisfy this equilibrium condition.

The exponential utility functions that we are using here have the remarkable property that an individual's marginal utility for a small increase in income in any state is proportional to his utility in that state. This mathematical fact leads the definition of adjusted probabilities which is shown in cells C16:F16 of Figure 7. For each state, the adjusted probability is defined to be the statistical probability multiplied by the utility value of the returns from the optimal investment portfolio, divided by the expected utility of this optimal investment. The adjusted probability of State 1 is computed in cell C16 by the formula

$$=C5*C11/ \$A\$12$$

This formula is copied from cell C16 to cells C16:F16, to compute the adjusted probabilities of all the possible states. Notice that these adjusted probabilities sum to one (because we divided by the EU), and they are all positive numbers (because the negative sign in the utility is canceled by the negative sign in the EU). These adjusted probabilities are the same for all investors (because their different optimal investments counterbalance their different risk tolerances in the utility function, as we noted in the preceding paragraph).

Now consider any other financial asset from which the returns next year will depend on the state. An example of returns next year per share of some stock X is shown in cells C19:F19 of Figure 7. The current value of this asset is computed in cell A20 by the formula

$$=\text{SUMPRODUCT}(C19:F19,\$C\$16:\$F\$16)/\$A\$1$$

That is, the current value of the asset is its adjusted-expected returns next year (computed with the adjusted probabilities rather than the statistical probabilities) divided by the risk-free return ratio.

When the price of an asset is equal to the value computed as above, individuals with optimal investments in the market portfolio will not want to sell or buy additional units of this asset. To demonstrate this result, the range A21:F26 in Figure 7 shows how to compute an individual's net demand for additional shares of this stock X. In these computations, the price of stock X is taken from cell A21, and the individual's net demand for additional shares is computed by Solver in cell A22. The problem for Solver here is to change the quantity in cell A22 so as to maximize the certainty equivalent in cell A26 (which takes account both the investment in the market portfolio and the net additional investment in this stock). If the investment in the market portfolio (cell A9) has already been set to maximize the certainty equivalent in cell A13 before any additional purchases of this stock are allowed, then the new optimal quantity of additional purchases in cell A22 will be 0 if and only if the price of the asset in cell A21 is equal to the computed value in cell A20. If the price in cell A21 is lower than the computed value in cell A20, then every investor will want to buy additional shares of this stock. If the price in cell A21 is higher than the computed value in cell A20, then every investor will want to sell off some shares of this stock. Thus, in a market equilibrium where no one wants to make any further investments or disinvestments, each asset's price must equal this computed value, which is the

adjusted-expected value divided by the risk-free return ratio.

Notice that, for this example, the adjusted-probabilities in this asset pricing model (C16:F16 in Figure 7) are very similar to the adjusted probabilities in the CAPM (C13:F13 in Figure 3). The computed value of this stock X in cell A20 of Figure 7 is almost the same as the CAPM value in cells A19 and A20 of Figure 3.

Figure 8 illustrates how simulation can be used to apply this asset pricing model. After generating a table of simulated returns from the market portfolio and this project, we ask Solver to find the optimal investment in the market portfolio for a constant-risk-tolerant investor. The adjusted-expected return from the project is estimated in cell C10 by averaging the simulated project returns multiplied by an adjustment factor (utility of the market portfolio in this simulated outcome, divided by average utility). The current value of the project (in cell C11) is this adjusted-expected value of returns divided by the risk-free return ratio.

(A note of comparison between Figures 6 and 8: In Figure 6, we used special properties of Brownian motion to find the arbitrage-adjusted probability distribution before simulating, and then we estimated the option's value by simulation data that was drawn from this adjusted probability distribution. Here in Figure 8, the simulation data is drawn from the correct statistical probability distribution, and so the data is reweighted using the column of adjustment factors.)

	A	B	C	D	E	F	G	
1	\$Returns next year per \$1 now in the market portfolio are Lognormal							
2	with mean=	1.14	and stdev=	0.19				
3								
4	The return ratio on risk-free debt is				1.1			
5								
6	What is the current value of an project yielding							
7	\$returns next year drawn from a Normal distribution							
8	with mean=	2000000	, stdev=	1000000	, and			
9	correlation	0.6	with the market portfolio?					
10	AdjustedE		1898370					
11	CurrentValue		1725791		With RT	EU		
12					2000	-0.9730		
13	corands	0.808989	0.05171663		invest in mkt	CE		
14	Simulated returns				2240	54.71		
15		Market	Project		and get			
16	SimTable	1.299563	371567.926		utility:	AdjustmentFactor		
17	0.00	0.79485	2178436.04		-1.4075	1.4465		
18	0.01	1.007985	2956863.46		-1.1086	1.1393		
115	0.98	1.069513	2712748.26		-1.0347	1.0634		
116	0.99	0.85522	1237720.66		-1.3154	1.3519		
117								
118	FORMULAS FROM RANGE A1:G17							
119	B13:C13. {=CORAND(B9)}							
120	B16. =LNORMINV(B13,B2,D2)							
121	C16. =NORMINV(C13,B8,D8)							
122	E17. =UTIL(\$E\$14*(B17-\$E\$4),\$E\$12)							
123	E17 copied to E17:E116							
124	F12. =AVERAGE(E17:E116)							
125	F14. =UINV(F12,E12)							
126	G17. =E17/\$F\$12							
127	G17 copied to G17:G116							
128	C10. =SUMPRODUCT(C17:C116,G17:G116)/COUNT(C17:C116)							
129	C11. =C10/E4							
130								
131	SOLVER: maximize F14 by changing E14.							

Figure 8. Evaluating a project with the constant-risk-tolerance asset pricing model.

## BROWNIAN MOTION EXERCISE:

PG Corporation stock is currently selling for \$140 per share. In recent years, annual logarithmic growth rates of PG stock have had an average of 0.18 and a standard deviation of 0.14.

So suppose that the future growth of PG stock value in PG will be a log-Brownian motion with annual log-drift 0.18, and log-volatility of 0.14.

Suppose that risk-free bonds offer an annual return ratio of 1.062, which corresponds to a logarithmic rate of return equal of  $\text{LN}(1.062) = 0.06$ .

Consider an option to buy one share of PG stock 2 years from now for \$180.

- (a) Assuming the statistical probabilities described above, estimate the probability that this option will have a positive value 2 years from now, and estimate the expected monetary value of this option 2 years from now.
- (b) The option may be exercised at any time in the next two years, but financial theorists have argued that such options should always be held to the latest exercise date. Assuming the statistical probabilities described above, estimate the probability of the event that PG stock will be below \$180 two years from now but PG stock will have closed above \$180 in at least one of the next 24 months.
- (c) Estimate the Black-Scholes value of this option now, using our technique of simulation with the arbitrage-adjusted probability distribution.
- (d) Estimate the Black-Scholes value now of a certificate that can be redeemed in two years for an amount equal to the highest monthly closing price of PG stock over each of the next 24 months.